<commit_before>{
  lib,
  buildGoModule,
  fetchFromGitHub,
  runCommand,
  installShellFiles,
}: let
  buildDate = "2022-08-15T21:36:49";
  buildCommit = "2<PASSWORD>";
in
  with lib;
    buildGoModule rec {
      pname = "doggo";
      version = "0.5.4";
      vendorSha256 = "sha256-pyzu89HDFrMQqYJZC2vdqzOc6PiAbqhaTgYakmN0qj8=";
      src = fetchFromGitHub {
        owner = "mr-karan";
        repo = "doggo";
        rev = "v0.5.4";
        sha256 = "sha256-6jNs8vigrwKk47Voe42J9QYMTP7KnNAtJ5vFZTUW680=";
      };

      nativeBuildInputs = [installShellFiles];
      subPackages = ["cmd/doggo"];

      ldflags = [
        "-w -s"
        "-X main.buildVersion=${buildCommit}"
        "-X main.buildDate=${buildDate}"
      ];

      postInstall = ''
        installShellCompletion --cmd doggo --fish --name doggo.fish completions/doggo.fish
        installShellCompletion --cmd doggo --zsh --name _doggo completions/doggo.zsh
      '';

      meta = {
        homepage = "https://github.com/mr-karan/doggo";
        description = "Command-line DNS Client for Humans. Written in Golang";
        longDescription = ''
          doggo is a modern command-line DNS client (like dig) written in Golang.
          It outputs information in a neat concise manner and supports protocols like DoH, DoT, DoQ, and DNSCrypt as well
        '';
        license = licenses.gpl3Only;
        platforms = platforms.linux;
        maintainers = with maintainers; [georgesalkhouri];
      };
    }
<commit_msg>Apply suggestions from code review<commit_after>{ lib
, buildGoModule
, fetchFromGitHub
, installShellFiles
}:

buildGoModule rec {
  pname = "doggo";
  version = "0.5.4";

  src = fetchFromGitHub {
    owner = "mr-karan";
    repo = pname;
    rev = "v${version}";
    sha256 = "sha256-6jNs8vigrwKk47Voe42J9QYMTP7KnNAtJ5vFZTUW680=";
  };

  vendorSha256 = "sha256-pyzu89HDFrMQqYJZC2vdqzOc6PiAbqhaTgYakmN0qj8=";
  nativeBuildInputs = [ installShellFiles ];
  subPackages = [ "cmd/doggo" ];

  ldflags = [
    "-w -s"
    "-X main.buildVersion=v${version}"
  ];

  postInstall = ''
    installShellCompletion --cmd doggo \
      --fish --name doggo.fish completions/doggo.fish \
      --zsh --name _doggo completions/doggo.zsh
  '';

  meta = with lib; {
    homepage = "https://github.com/mr-karan/doggo";
    description = "Command-line DNS Client for Humans. Written in Golang";
    longDescription = ''
      doggo is a modern command-line DNS client (like dig) written in Golang.
      It outputs information in a neat concise manner and supports protocols like DoH, DoT, DoQ, and DNSCrypt as well
    '';
    license = licenses.gpl3Only;
    platforms = platforms.linux;
    maintainers = with maintainers; [ georgesalkhouri ];
  };
}

{'fraction_non_alphanumeric': 0.11642335766423358, 'fraction_numerical': 0.031386861313868614, 'mean_word_length': 2.704054054054054, 'pattern_counts': {'":': 0, '<': 4, '<?xml version=': 0, '>': 4, 'https://': 2, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6874201', 'n_tokens_mistral': 1007, 'n_tokens_neox': 944, 'n_words': 212}